7

15.7.2
Special symbols that can be used instead of values

These matching mechanisms can be used to characterize a set of values. 

Syntactical Structure
omit |

"(" { (TemplateInstance | all from TemplateInstance) [","] } ")" |

complement "(" { (TemplateInstance | all from TemplateInstance) [","] } ")" |

"?" |

"
*" |

"(" ( ConstantExpression | -infinity ) ".." ( ConstantExpression | infinity ) ")" |

superset "(" { (TemplateInstance | all from TemplateInstance) [","] } ")" |

subset "(" { (TemplateInstance | all from TemplateInstance) [","] } ")" |

pattern Cstring

Semantic Description
The matching mechanisms for special symbols that can be used instead of values are:

· omit: the optional field, in which it is used, is not present;

· (…): a list of values or templates;

· complement (…): complement of a list of values or templates;

· ?: wildcard for any value;

· *: wildcard for any value or no value at all, i.e. the field is not present;

· (lowerBound .. upperBound): a range of integer or float values between and including the lower- and upper bounds;

· superset: at least all of the elements listed, i.e. possibly more;

· subset: at most the elements listed, i.e. possibly less;

· pattern: a charstring or universal charstring that matches this format.

Some of the above mentioned matching mechanisms (list, complemented list, subset, superset) can reference other templates directly or include a complete content of another template using the all from clause. For further details please refer to annex B.

Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:

a) See the restrictions given in table 11 and in annex B.

Examples

MyPCO.receive (integer:complement(1, 2, 3));

15.7.3
Special symbols that can be used inside values

These matching mechanisms allow to characterize value sets by varying values inside. 

Syntactical Structure
…
"?"… |

…
"*"… |

…
permutation "(" { ( TemplateBody | "?" | "*" | all from TemplateInstance )[","] } ")"…

Semantic Description
The matching mechanisms for special symbols that can be used inside values are:

· ?: wildcard for any single element in a string, array, record of or set of;
· *: wildcard for any number of consecutive elements in a string, array, record of or set of, or no element at all (i.e. an omitted element);

· permutation: all of the elements listed but in an arbitrary order (note, that ? and * are also allowed as elements of the permutation list and all elements of another template can be added to permutation using the all from clause).

For further details please refer to annex B.

Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:

a) See the restrictions given in table 11 and in annex B.

Examples

template bitstring b := '10???'B; 

// where each "?" may either be 0 or 1 


type record of integer RI;


template RI ri := {1, ?, 3}



// where ? may be any integer value

A.1.6
TTCN-3 syntax BNF productions

A.1.6.1.3
Template definitions

 AUTONUM  Complement ::= ComplementKeyword "(" TemplateListItem {"," TemplateListItem }  

                    ")" 

 AUTONUM  ComplementKeyword ::= "complement" 

 AUTONUM  SubsetMatch ::= SubsetKeyword TemplateList


 AUTONUM  SubsetKeyword ::= "subset" 

 AUTONUM  SupersetMatch ::= SupersetKeyword TemplateList
 AUTONUM  SupersetKeyword ::= "superset" 

 AUTONUM  PermutationMatch ::= PermutationKeyword PermutationList 

 AUTONUM  PermutationKeyword ::= "permutation" 

 AUTONUM  PermutationList ::= "(" PermutationListItem {"," PermutationListItem} ")" 
PermutationListItem ::= (TemplateBody | AllElementsFrom)
/* STATIC SEMANTICS: Restrictions on the content of TemplateBody are given in clause B.1.3.3. */ 

 AUTONUM  AnyValue ::= "?" 

 AUTONUM  AnyOrOmit ::= "*" 

 AUTONUM  TemplateList ::= "(" (TemplateBody {"," TemplateListItem}+) |





(AllElementsFrom {"," TemplateListItem}) ")" 

12A. TemplateListItem ::= (TemplateBody | AllElementsFrom)
12B. AllElementsFrom ::= AllKeyword FromKeyword TemplateBody
/* STATIC SEMANTICS: TemplateBody shall resolve into a value or template of a record of or set of type */
B.1.2.1
Template list

Template lists specify lists of acceptable values. It can be used on values of all types. A template list may also contain templates. 

A template field that uses a template list matches the corresponding field if, and only if, the field value matches any one of the values or templates in the template list. Each value or template in the template list shall be of the type declared for the template field in which this mechanism is used.

EXAMPLE 1:

template MyMessage MyTemplate:= 


{ 




field1 := (2,4,6),



// list of integer values 



field2 := ("String1", "String2"),
// list of charstring values 



:



:


}
Beside specifying individual values, it is also possible to add all elements of an existing record of or set of template into template lists using an all from clause.

Restrictions
a)
The type of the template list and the member type of the record of or set of type associated with the template in the all from claused shall be compatible.

b)
The template in the all from clause as a whole shall not resolve into a matching mechanism (i.e. its elements may contain any of the matching mechanisms or matching attributes with the exception of those described in the following restriction).
c)
Individual fields of the template in the all from clause shall not resolve to any of the following matching mechanisms: AnyElementsOrNone, permutation.

EXAMPLE 2:

type record of integer RoI;


template RoI t_RoI1 := {1, 2, (6..9)};

template RoI t_RoI2 := {1, *, 3};


template integer t_i1 := (all from t_RoI1, 100); // will result into (1, 2, (6..9), 100)


template integer t_i2 := (0, all from t_RoI2); // will produce an error, because t_RoI2















// contains AnyElementsOrNone

template RoI t_RoI3 := (all from t_RoI1); // will produce an error, because member type of








// t_RoI1 (integer) is not compatible with the list template type (RoI)


template RoI t_RoI4 := ?;


template RoI t_RoI5 := (all from t_RoI4); // will produce an error, because t_RoI4 resolves 











// into a matching mechanism
B.1.2.2
Complemented template list

The keyword complement denotes a list of values that will not be accepted as values (i.e. it is the complement of a template list). It can be used on all values of all types. A complemented value list may also contain templates.

Each value or template in the list shall be of the type declared for the template field in which the complement is used. 

A template field that uses complement matches the corresponding field if and only if the field does not match any of the values or templates listed in the template list. The template list may be a single value, of course.

EXAMPLE 1:


template MyMessage MyTemplate:= 


{




complement (1,3,5),
// list of unacceptable integer values 



:



field3 not(true)

// will match false 



:


}

Beside specifying individual values, it is possible to add all elements of an existing record of or set of template into complement template lists using an all from clause.

Restrictions
a)
The type of the complemented template list and the member type of the record of or set of type associated with the template in the all from claused shall be compatible.

b)
The template in the all from clause as a whole shall not resolve into a matching mechanism (i.e. its elementsmay contain any of the matching mechanisms or matching attributes with the exception of those described in the following restriction).
c)
Individual fields of the template in the all from clause shall not resolve to any of the following matching mechanisms: AnyElementsOrNone, permutation.

EXAMPLE 2:

type record of integer RoI;


template RoI t_RoI1 := {1, 2, (6..9)};

template RoI t_RoI2 := {1, *, 3};


template integer t_i1 := complement(all from t_RoI1, 100); 


// will result into (1, 2, (6..9), 100)


template integer t_i2 := complement(0, all from t_RoI2); 


// will produce an error, because t_RoI2 contains AnyElementsOrNone

template RoI t_RoI3 := complement(all from t_RoI1); 



// will produce an error, because member type of t_RoI1 (integer) is not compatible 




// with the complemented list template type (RoI)


template RoI t_RoI4 := ?;


template RoI t_RoI5 := complement (all from t_RoI4); 



// will produce an error, because t_RoI4 resolves into a matching mechanism
B.1.2.6
SuperSet

SuperSet is an operation for matching that shall be used only on values of set of types. SuperSet is denoted by the keyword superset. SuperSet matches a set of values if, and only if, the set of values contains at least all of the elements defined within the SuperSet, and may contain more. This argument may contain templates (including template variables) and matching mechanisms, with the exceptions given in the Restrictions paragraph of this clause. However, the length matching attribute may be attached to the SuperSet itself, in which case the minimal length allowed by the length attribute shall not be less than the number of the elements in the SuperSet.

EXAMPLE 1:


type set of integer MySetOfType (0 .. 10);


template MySetOfType MyTemplate1 := superset (1, 2, 3);


// matches any sequence of integers which contains at least one occurrences of the numbers


// 1, 2 and 3 in any order and position


template MySetOfType MyTemplate2_AnyValue := superset (1, 2, ?);


// matches any sequence of integers which contains at least one occurrences of the numbers


// 1, 2 and at least one more valid integer value (i.e. between 0 and 10, inclusively), in any


// order and position


template MySetOfType MyTemplate3 := superset (1, 2, (3, 4));


// matches any sequence of integers which contains at least one occurrences of the numbers


// 1, 2 and a number with the value 3 or 4, in any order and position


template MySetOfType MyTemplate4 := superset (1, 2, complement(3, 4));


// any sequence of integers matches which contains at least one occurrences of the numbers


// 1, 2 and a valid integer value which is not 3 or 4, in any order and position


template MySetOfType MyTemplate6 := superset (1, 2, 3) length (7);


// matches any sequence of 7 integers which contains at least one occurrences of the numbers


// 1, 2 and 3 in any order and position


template MySetOfType MyTemplate7 := superset (1, 2, ?) length (7 .. infinity);


// matches any sequence of at least 7 integers which contains at least one occurrences of the


// numbers 1, 2 and 3 in any order and position


template MySetOfType MyTemplate8 := superset (1, 2, 3) length (2 .. 7);


// causes an error, the lower bound of the length attribute contradicts to the minimum number


// of elements imposed by the superset argument

Beside specifying individual values, it is possible to add all elements of an existing record of or set of template into supersets using an all from clause.

Restrictions
a)
Individual members of the SuperSet’s argument shall be of the type replicated by the set of.
b)
The member type of the set of type associated with the superset template and the member type of the record of or set of type associated with the template in the all from claused shall be compatible.

c)
The template in the all from clause as a whole shall not resolve into a matching mechanism (i.e. its elementsmay contain any of the matching mechanisms or matching attributes with the exception of those described in the following restriction).
d)
The individual members of the SuperSet’s argument and the elements of the template in the all from clause shall not be the matching mechanisms omit, superset, subset and the matching attributes (length restriction and ifpresent). In addition, the individual members shall not resolve to AnyValueOrNone and individual elements of the template in an all from clause shall not resolve to AnyElementsOrNone or permutation.

EXAMPLE 2:

type record of integer RoI;


type set of integer SoI;

template RoI t_RoI1 := {1, 2, ?};

template SoI t_SoI1 := superset(all from t_RoI1); 


// will result into superset(1, 2, ?)
B.1.2.7
SubSet

SubSet is an operation for matching that can be used only on values of set of types. SubSet is denoted by the keyword subset. SubSet matches a set of values if, and only if, the set of values contains only elements defined within the SubSet, and may contain less. This argument may contain templates (including template variables) and matching mechanisms, with the exceptions given in the Restrictions paragraph of this clause. However, the length matching attribute may be attached to the SubSet itself, in which case the maximum length allowed by the length attribute shall not exceed the number of the elements in the SubSet.

EXAMPLE 1:


template MySetOfType MyTemplate1:= subset (1, 2, 3);


// matches any sequence of integers which contains zero or one occurrences of the numbers


// 1, 2 and 3 in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, ?);


// matches any sequence of integers which contains zero or one occurrences of the numbers


// 1, 2 and a valid integer value (i.e. between 0 and 10, inclusive) in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, (3, 4));


// matches any sequence of integers which contains zero or one occurrences of the numbers


// 1, 2 and one of the numbers 3 or 4, in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, complement (3, 4));


// matches any sequence of integers which contains zero or one occurrences of the numbers


// 1, 2 and a valid integer number which is not 3 or 4, in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, 3) length (2);


// matches any sequence of two integers which contains zero or one occurrences of


// the numbers 1, 2 and 3, in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, ?) length (0 .. 2);


// matches any sequence of zero, one or two integers which contains zero or one occurrences of


// the numbers 1, 2 and of a valid integer value, in any order and position


template MySetOfType MyTemplate1:= subset (1, 2, 3) length (0 .. 4);


// causes an error, the upper bound of length attribute contradicts to the maximum number of


// elements imposed by the subset argument

Beside  specifying individual values, it is possible to add all elements of an existing record of or set of template into subsets using an all from clause.

Restrictions
a)
Individual members of the SubSet’s argument shall be of the type replicated by the set of.
a)
The member type of the set of type associated with the subset and the member type of the record of or set of type associated with the template in the all from claused shall be compatible.

b)
The template in the all from clause as a whole shall not resolve into a matching mechanism (i.e. its elementsmay contain any of the matching mechanisms or matching attributes with the exception of those described in the following restriction).
c)
The individual members of the SubSet’s argument and the elements of the template in the all from clause shall not be the matching mechanisms omit, superset, subset and the matching attributes (length restriction and ifpresent). In addition, individual members shall not resolve to AnyValueOrNone and individual fields of the template in the all from clause shall not resolve to AnyElementsOrNone or permutation.

EXAMPLE 2:

type record of integer RoI;


type set of integer SoI;

template RoI t_RoI1 := {1, 2, ?};

template SoI t_SoI1 := subset(all from t_RoI1); 


// will result into subset(1, 2, ?)
B.1.3.3
Permutation

Permutation is an operation for matching that shall be used only on values of record of types. Permutation is denoted by the keyword permutation. Permutation elements shall obey the constraints given in the Restrictions paragraph of this clause. 
A permutation without AnyElementsOrNone in place of a single record of element means that any series of elements is acceptable provided that there is a one to one mapping between elements in the record of and in the permutation list such that each element matches its corresponding element in the permutation list. 

AnyElementsOrNone used inside permutation (directly or via reference) replaces none or any number of elements within the segment of the record of value matched by permutation. The permutation matching is successful, if a subset of the elements in the record of matches the permutation list without the AnyElementsOrNone. If both permutation and AnyElementsOrNone are used in a record of template, they shall be evaluated jointly.

NOTE 1:
AnyElementsOrNone used inside permutation has a different effect as AnyElementsOrNone used in conjunction with permutation as in the latter AnyElementsOrNone replaces consecutive elements only. For example, {permutation(1,2,*)} is equivalent to ({*,1,*,2,*},{*,2,*,1,*}), while {permutation(1,2),*} is equivalent to ({1,2,*},{2,1,*}).

NOTE 2:
When AnyElementsOrNone is inside a permutation, a length attribute may be applied to AnyElementsOrNone to restrict the number of elements matched by AnyElementsOrNone (see also clause B.1.4.1). 

EXAMPLE 1:


type record of integer MySequenceOfType;


template MySequenceOfType MyTemplate1 := { permutation ( 1, 2, 3 ), 5 };


// matches any of the following sequences of 4 integers: 1,2,3,5; 1,3,2,5; 2,1,3,5;


// 2,3,1,5; 3,1,2,5; or 3,2,1,5

template MySequenceOfType MyTemplate2 := { permutation ( 1, 2, ? ), 5 };


// matches any sequence of 4 integers that ends with 5 and contains 1 and 2 at least once in


// other positions


template MySequenceOfType MyTemplate3 := { permutation ( 1, 2, 3 ), * };


// matches any sequence of integers starting with 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1


template MySequenceOfType MyTemplate4 := { *, permutation ( 1, 2, 3 )};


// matches any sequence of integers ending with 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1


template MySequenceOfType MyTemplate5 := { *, permutation ( 1, 2, 3 ),* };


// matches any sequence of integers containing any of the following substrings at any position:


// 1,2,3; 1,3,2; 2,1,3; 2,3,1; 3,1,2 or 3,2,1 


template MySequenceOfType MyTemplate6 := { permutation ( 1, 2, * ), 5 };


// matches any sequence of integers that ends with 5 and containing 1 and 2 at least once in


// other positions


template MySequenceOfType MyTemplate7 := { permutation ( 1, 2, 3 ), * length (0..5)};


// matches any sequence of three to eight integers starting with 1,2,3; 1,3,2; 2,1,3; 2,3,1;


// 3,1,2 or 3,2,1


template integer MyInt1 := (1,2,3);

template integer MyInt2 := (1,2,?);

template integer MyInt3 := ?;

template integer MyInt4 := *;

template MySequenceOfType MyTemplate10 := { permutation (MyInt1, 2, 3 ), 5 };


// matches any of the sequences of 4 integers:


//

1,3,2,5; 2,1,3,5; 2,3,1,5; 3,1,2,5; or 3,2,1,5;

//

2,3,2,5; 2,2,3,5; 2,3,2,5; 3,2,2,5; or 3,2,2,5;

//

3,3,2,5; 2,3,3,5; 2,3,3,5; 3,3,2,5; or 3,2,3,5;

template MySequenceOfType MyTemplate11 := { permutation (MyInt2, 2, 3 ), 5 };


// matches any sequence of 4 integers that ends with 5 and contains 2 and 3 at least once in


// other positions

template MySequenceOfType MyTemplate12 := { permutation (MyInt3, 2, 3 ), 5 };


// matches any sequence of 4 integers that ends with 5 and contains 2 and 3 at least once in


// other positions

template MySequenceOfType MyTemplate13 := { permutation (MyInt4, 2, 3 ), 5 };


// matches any sequence of integers that ends with 5 and containing 2 and 3 at least once in


// other positions


template MySequenceOfType MyTemplate14 := { permutation (MyInt3, 2, ? ), 5 };


// matches any sequence of 4 integers that ends with 5 and contains 2 at least once in


// other positions


template MySequenceOfType MyTemplate15 := { permutation (MyInt4, 2, * ), 5 };


// matches any sequence of integers that ends with 5 and contains 2 at least once in


// other positions

Beside of specifying all individual values, it is possible to add all elements of an existing record of or set of template into permutations using an all from clause.

Restrictions
a)
Each individual member listed in the permutation shall be of the type replicated by the record of type.
a)
The member type of the record of type associated with the template that contains permutation and the member type of the record of or set of type associated with the template in the all from claused shall be compatible.

b)
The template in the all from clause as a whole shall not resolve into a metchaning mechanism (i.e. its fields may contain any of the matching mechanisms or matching attributes with the exception of those described in the following restriction).
c)
Individual members of a permutation and elements of the template in the all from clause shall only be Expressions, templates, and the AnyElement and AnyElementsOrNone matching mechanisms.
EXAMPLE 2:

type record of integer RoI;


template RoI t_RoI1 := {1, 2, ?};

template RoI t_RoI2 := {permutation(0, all from t_RoI1), 4, 5}; 


// will result into {permutation(0, 1, 2, ?), 4, 5}
ETSI

